
Installation
    


  
    
  
  Prerequisites


Make sure you have a running phoenix application with Tailwind CSS and daisyUI installed.
Important
Backpex currently only supports daisyUI light mode.

Backpex currently depends on Ecto. Therefore there has to be a running
ecto repository.

  
    
  
  Add Backpex to list of dependencies


In your mix.exs:
defp deps do
  [
    ...
    {:backpex, path: "backpex_path"}
  ]
end

  
    
  
  Add Backpex files to tailwind content


In your tailwind.config.js:
content: ['./path_to_deps/backpex/**/*.*ex']

  
    
  
  Setup formatter


We recommend to add Backpex to the list of dependencies in your .formatter.exs.
# my_app/.formatter.exs
[
  import_deps: [:backpex]
]

  
    
  
  Create layout


Backpex does not ship a predefined layout by default. Instead it exposes components you
may use to build your own layout. You could also partly use Backpex components. For example
you are able to define your own sidebar navigation in our app shell component.
Place the layout file in your lib/myapp_web/templates/layout directory.
You can find all Backpex components in the lib/backpex/components directory.
If you do not want to put effort into creating your own layout, feel free to use this layout (admin.html.heex) as a starting point:
The Backpex.HTML.Layout.app_shell component accepts a boolean fluid to determine if a LiveResource should be rendered full width. Our default is the definition of the fluid? option in a LiveResource, but feel free to change this behavior in your layout.

<Backpex.HTML.Layout.app_shell fluid={@fluid?}>
  <:topbar>
    <Backpex.HTML.Layout.topbar_branding />
    <Backpex.HTML.Layout.topbar_dropdown class="rounded-md p-1 hover:bg-gray-100">
      <:label>
        <Heroicons.user class="h-8 w-8" />
      </:label>
      <div class="py-1">
        <.link navigate="/" class="my-1 flex justify-between px-4 py-2 text-sm text-red-600 hover:bg-gray-100">
          <p>Logout</p>
          <Heroicons.arrow_right_on_rectangle class="h-5 w-5" />
        </.link>
      </div>
    </Backpex.HTML.Layout.topbar_dropdown>
  </:topbar>
  <:sidebar>
    <Backpex.HTML.Layout.sidebar_item current_url={@current_url} navigate="/">
      <Heroicons.user class="h-5 w-5" /> Your Resource
    </Backpex.HTML.Layout.sidebar_item>
    <Backpex.HTML.Layout.sidebar_section id="your_section">
      <:label>Your Sidebar Section</:label>
      <Backpex.HTML.Layout.sidebar_item current_url={@current_url} navigate="/">
        <Heroicons.book_open class="h-5 w-5" /> Your nested Resource
      </Backpex.HTML.Layout.sidebar_item>
    </Backpex.HTML.Layout.sidebar_section>
  </:sidebar>
  <Backpex.HTML.Layout.flash_messages flash={@flash} />
  <%= @inner_content %>
</Backpex.HTML.Layout.app_shell>
You can now create and configure the corresponding resources.


  

    
Translations
    

You may configure translator functions in your application config:
config :backpex, :translator_function, {MyAppWeb.Helpers, :translate_backpex}

config :backpex, :error_translator_function, {MyAppWeb.ErrorHelpers, :translate_error}
The first one is being used to translate general strings. The second one is being used to translate
(changeset) errors.

  
    
  
  Using Gettext


If you want to use Gettext, the translator functions should look like this:
# MyAppWeb.Helpers

def translate_backpex({msg, opts}) do
  if count = opts[:count] do
    Gettext.dngettext(MyAppWeb.Gettext, "backpex", msg, msg, count, opts)
  else
    Gettext.dgettext(MyAppWeb.Gettext, "backpex", msg, opts)
  end
end

# MyAppWeb.ErrorHelpers

def translate_error({msg, opts}) do
  if count = opts[:count] do
    Gettext.dngettext(DemoWeb.Gettext, "errors", msg, msg, count, opts)
  else
    Gettext.dgettext(DemoWeb.Gettext, "errors", msg, opts)
  end
end
Store this example template as priv/gettext/backpex.pot. It contains all translations used by Backpex.
## This file is a PO Template file.
msgid ""
msgstr ""

msgid "New %{resource}"
msgstr ""

msgid "Edit %{resource}"
msgstr ""

msgid "Delete"
msgstr ""

msgid "Edit"
msgstr ""

msgid "Save"
msgstr ""

msgid "Cancel"
msgstr ""

msgid "Search"
msgstr ""

msgid "Filters"
msgstr ""

msgid "Are you sure you want to delete %{count} items?"
msgstr ""

msgid "Are you sure you want to delete the item?"
msgstr ""

msgid "New %{resource} has been created successfully."
msgstr ""

msgid "%{resource} has been edited successfully."
msgstr ""

msgid "%{resource} has been deleted successfully."
msgstr ""

msgid "%{count} %{resources} have been deleted successfully."
msgstr ""

msgid "An error occurred while deleting the %{resource}!"
msgstr ""

msgid "An error occurred while deleting %{count} %{resources}!"
msgstr ""

msgid "The item is used elsewhere."
msgstr ""

msgid "The items are used elsewhere."
msgstr ""

msgid "You are not allowed to do this!"
msgstr ""

msgid "Upload a file"
msgstr ""

msgid "or drag and drop"
msgstr ""

msgid "too large"
msgstr ""

msgid "too many files"
msgstr ""

msgid "unacceptable file type"
msgstr ""

msgid "clear"
msgstr ""

msgid "No %{resources} found"
msgstr ""

msgid "Try a different search term."
msgstr ""

msgid "Try a different filter setting or clear all filters."
msgstr ""

msgid "Get started by creating new %{resources}."
msgstr ""

msgid "Select options..."
msgstr ""

msgid "No options found"
msgstr ""

msgid "Select all"
msgstr ""

msgid "Deselect all"
msgstr ""

msgid "Show more"
msgstr ""

msgid "Items %{from} to %{to}"
msgstr ""

msgid "total"
msgstr ""

msgid "There are errors in the form."
msgstr ""

msgid "Add entry"
msgstr ""

msgid "Apply"
msgstr ""

msgid "Attach %{resource}"
msgstr ""

msgid "Choose %{resource} ..."
msgstr ""

msgid "Toggle metrics"
msgstr ""

msgid "Toggle columns"
msgstr ""

msgid "Select all items"
msgstr ""

msgid "Close alert"
msgstr ""

msgid "Select item with id: %{id}"
msgstr ""

msgid "Clear %{name} filter"
msgstr ""

msgid "Edit relation with index %{index}"
msgstr ""

msgid "Detach relation with index %{index}"
msgstr ""

msgid "Error in relation with index %{index}"
msgstr ""

msgid "Close modal"
msgstr ""

msgid "Unselect %{label}"
msgstr ""

msgid "selected"
msgstr ""


  

    
Full-Text Search
    

Backpex allows you to perform full-text searches on resources. It uses the built-in PostgreSQL full-text search functionality.

  
    
  
  Create a Generated Column


Backpex forces you to create a generated column to use the full-text search functionality. It must contain a tsvector that is generated from all the columns that you want to be considered when searching. You are free to choose a name for this column.
Below is an example of a generated column for a movie review resource with a title column and an overview column. Both columns should be searchable.
# in the database up migration

execute("""
ALTER TABLE film_reviews
  ADD COLUMN generated_tsvector tsvector
  GENERATED ALWAYS AS (
    to_tsvector('english', coalesce(title, '') || ' ' || coalesce(overview, ''))
  ) STORED;
""")
You can also concat multiple tsvectors in the generated column. This is useful if the table contains data in
different languages. We recommend that you specify the language when using the to_tsvector function. Otherwise
the default language will be used.

  
    
  
  Create an Index


To increase the speed of full-text searches, especially for resources with large amounts of data, you should create
an index on the generated column created in the previous step.
We strongly recommend that you use a GIN index, as it makes full-text searches really fast.
The disadvantage is that a GIN index takes up a lot of disk space, so if you are limited in disk space,
feel free to use a GiST index instead.
# in the database up migration

execute("""
CREATE INDEX film_reviews_search_idx ON film_reviews USING GIN(generated_tsvector);
""")
Important
Note that you must explicitly define up and down migrations. Otherwise, the index cannot be dropped.

# in the database down migration

execute("""
DROP INDEX film_reviews_search_idx;
""")

drop table(:film_reviews)
To enable full-text search, you need to specify the name of the generated column in the live resource of the
corresponding resource.
# in the live resource

use Backpex.LiveResource,
  full_text_search: :generated_tsvector


  

    
Upgrading to v0.3
    


  
    
  
  Bump Your Deps


Update Backpex to the latest version:
  defp deps do
    [
      {:backpex, "~> 0.3.0"}
    ]
  end

  
    
  
  Update calls to Backpex.Resource


We have updated certain functions in Backpex.Resource to make them more explicit and usable from outside. 
The following functions are affected:
	Backpex.Resource.get!/5 (get/4 before)
	Backpex.Resource.delete/3 (delete/2 before)
	Backpex.Resource.delete_all/4 (delete_all/2 before)
	Backpex.Resource.update/5 (update/2 before)
	Backpex.Resource.update_all/6 (update_all/4 before)
	Backpex.Resource.insert/5 (insert/2 before)

If you call one of these functions in your application, you will probably need to update the function call.
We added extensive documentation to the updated functions in Backpex.Resource.
See Pull Request #269 for more information.

  
    
  
  Update Upload callback functions


We have updated Backpex.Fields.Upload and improved uploads in Backpex:
	Uploaded and existing files are now included in the change to allow them to be validated in the changeset
	Uploads can be required
	Uploads are not removed until form is saved
	Form field errors are displayed for upload field

For these requirements we have simplified the functions in Backpex.FormComponent and adapted the validation flow to classic Phoenix applications.
We also updated the callback functions / options used for Backpex.Fields.Upload. Among other things, we have split consume/2 into several callbacks (put_upload_change/6 and consume_upload/6). This allows developers to put uploaded files to the change before consuming the uploads. This makes it possible to do validation on uploads (e.g. require files to be uploaded).
We have also rewritten all the upload documentation in Backpex.Fields.Upload. It contains full examples for single and multiple upload fields.
We recommend that you read the new upload documentation and adapt the callback functions in your application accordingly.
See Pull Request #269 for more information.


  

    
Upgrading to v0.2
    


  
    
  
  Bump Your Deps


Update Backpex to the latest version:
  defp deps do
    [
      {:backpex, "~> 0.2.0"}
    ]
  end

  
    
  
  Pass assigns to init_change functions


We change the arity of the init_change/0 function to init_change/1 for resource and item actions.
The param will be the assigns. This adds more ways to construct an initial change.
If you had such a function in your resource or item action:
@impl Backpex.ItemAction
def init_change() do
  # construct init change
end
You need to change it to this:
@impl Backpex.ItemAction
def init_change(_assigns) do
  # construct init change
end

  
    
  
  Change arity of changeset functions


See Pull Request.
We are introducing a new way to pass metadata to changeset functions. Previously, we supported changeset functions with
different arities to optionally pass the assigns and target name to changesets.
With this release, we require the arity of all changeset functions you provide to be 3.
The metadata, previously passed as additional parameters, is now passed as a keyword list to changesets as the third parameter.
Currently we pass the following metadata to changesets:
	:assigns - the assigns
	:target - the name of the form target that triggered the changeset call

When you previously had a LiveResource and Schema that looked like the following:
defmodule MyAppWeb.UserLive do
  use Backpex.LiveResource,
      ...
      update_changeset: &MyApp.User.changeset/2,
      create_changeset: &MyApp.User.changeset/2,
end


defmodule MyApp.User do
  ...

  def changeset(user, attrs) do
      ...
  end
end
You have to change it to this:
defmodule MyAppWeb.UserLive do
  use Backpex.LiveResource,
    ...
    update_changeset: &MyApp.User.changeset/3,
    create_changeset: &MyApp.User.changeset/3,
end


defmodule MyApp.User do
  ...

  def changeset(user, attrs, metadata) do
    # fetch assigns from metadata
    assigns = Keyword.get(metadata, :assigns)

    # fetch target from metadata
    assigns = Keyword.get(metadata, :target)

    ...
  end
end
This change applies to Resource and Item Actions as well.
For example:
defmodule MyApp.EmailResourceAction
  use Backpex.ResourceAction

  @impl Backpex.ResourceAction
  def changeset(change, attrs, _metadata \\ []) do
    ...
  end
end

  
    
  
  Update prompt option


With v0.2 we are updating the prompt option for fields. This affects to the following fields: BelongsTo, Select, HasMany, ManyToMany and MultiSelect. 
The prompt option can be raw text or a function that takes the assigns and returns text. 
We no longer support adding additional options.
If you previously had a field that looked like the following:
def fields do
  [
    user: %{
      module: Backpex.Fields.BelongsTo,
      label: "Author",
      prompt: [key: "Please select an author", disabled: true],
      ...
    },
  ]
end
You need to change it to the following:
def fields do
  [
    user: %{
      module: Backpex.Fields.BelongsTo,
      label: "Author",
      prompt: "Please select an author",
      ...
    },
  ]
end


  

    
Backpex 
    



      
Backpex provides an easy way to manage existing resources in your application.

    Backpex.CookieController - Backpex v0.3.0

    Backpex.Ecto.Amount.Type - Backpex v0.3.0
    
    

    


  
  

    
Backpex.Ecto.Amount.Type 
    



      
Provides a type for Ecto to store a amount.
The underlying data type should be an integer.

  
    
  
  Migration


create table(:my_table) do
  add :amount, :integer
end

  
    
  
  Schema


schema "my_table" do
  field :amount, Backpex.Ecto.Amount.Type
end

schema "my_table" do
  field :amount, Backpex.Ecto.Amount.Type, currency: :EUR, separator: ".", delimiter: ","
end

    Backpex.FormComponent - Backpex v0.3.0
    
    

    


  
  

    
Backpex.FormComponent 
    



      
The form live component.

    Backpex.InitAssigns - Backpex v0.3.0
    
    

    


  
  

    
Backpex.InitAssigns 
    



      
Ensures Backpex assigns are applied to all LiveViews attaching this hook.

    Backpex.LiveResource - Backpex v0.3.0
    
    

    


  
  

    
Backpex.LiveResource behaviour
    



      
LiveResource makes it easy to manage existing resources in your application.
It provides extensive configuration options in order to meet everyone's needs.
In connection with Backpex.Components you can build an individual admin dashboard
on top of your application in minutes.

  
    
  
  Example


Before you start make sure Backpex is properly configured.
Backpex.LiveResource is the module that will generate the corresponding LiveViews for the resource you configured.
You are required to set some general options to tell Backpex where to find the resource and what
changesets should be used. In addition you have to provide names and a list of fields.
A minimal configuration looks something like this:
defmodule MyAppWeb.UserLive do
  use Backpex.LiveResource,
    layout: {MyAppWeb.LayoutView, :admin},
    schema: MyApp.User,
    repo: MyApp.Repo,
    update_changeset: &MyApp.User.update_changeset/3,
    create_changeset: &MyApp.User.create_changeset/3

  @impl Backpex.LiveResource
  def singular_name(), do: "User"

  @impl Backpex.LiveResource
  def plural_name(), do: "Users"

  @impl Backpex.LiveResource
  def fields do
  [
    username: %{
      module: Backpex.Fields.Text,
      label: "Username"
    },
    first_name: %{
      module: Backpex.Fields.Text,
      label: "First Name"
    },
    last_name: %{
      module: Backpex.Fields.Text,
      label: "Last Name"
    },
  ]
end
You are also required to configure your router in order to serve the generated LiveViews:
defmodule MyAppWeb.Router
  import Backpex.Router

  scope "/admin", MyAppWeb do
    pipe_through :browser

    live_session :default, on_mount: Backpex.InitAssigns do
      live_resources("/users", UserLive)
    end
  end
end
use Backpex.LiveResource
When you use Backpex.LiveResource, the Backpex.LiveResource module will set @behavior Backpex.LiveResource. Additionally it will create a LiveView based on the given configuration in order to create fully functional index, show, new and edit views for a resource. It will also insert fallback functions that can be overridden.


  
    
  
  Define a resource


To explain configuration settings we created an event schema with a corresponding EventLive configuration file.
defmodule MyAppWeb.EventLive do
  alias MyApp.Event

  use Backpex.LiveResource,
    layout: {MyAppWeb.LayoutView, :admin}, # Specify the layout you created in the step before
    schema: Event, # Schema of the resource
    repo: MyApp.Repo, # Ecto repository
    update_changeset: &Event.update_changeset/3, # Changeset to be used for update queries
    create_changeset: &Event.create_changeset/3,  # Changeset to be used for create queries
    pubsub: Demo.PubSub, # PubSub name of the project.
    topic: "events", # The topic for PubSub
    event_prefix: "event_", # The event prefix for Pubsub, to differentiate between events of different resources when subscribed to multiple resources
    fluid?: true # Optional to define if your resource should be rendered full width. Depends on the your [layout configuration](installation.md)

  # Singular name to displayed on the resource page
  @impl Backpex.LiveResource
  def singular_name(), do: "Event"

  # Plural name to displayed on the resource page
  @impl Backpex.LiveResource
  def plural_name(), do: "Events"

  # Field configurations
  # Here can configure which fields of the schema should be displayed on your dashboard.
  # Backpex provides certain field modules that may be used for displaying the field on index and form views.
  # You may define you own module or overwrite the render functions in this configuration (`render` for index views
  # and `render_form` for form views).
  @impl Backpex.LiveResource
  def fields do
    [
      # The title field of our event schema
      title: %{
        module: Backpex.Fields.Text,
        label: "Title"
      },
      # The location field of our event schema. It should not be displayed on `:edit` view.
      location: %{
        module: Backpex.Fields.Text,
        label: "Location",
        except: [:edit]
      },
      # The location field of our event schema. We use the Backpex URL module in order to make the url clickable.
      # This field is only displayed on `:index` view.
      url: %{
        module: Backpex.Fields.URL,
        label: "Url",
        only: [:index]
      },
      # The begins_at field of our event schema. We provide or own render function to display this field on index views.
      # The value can be extracted from the assigns.
      begins_at: %{
        module: Backpex.Fields.Date,
        label: "Begins At",
        render: fn assigns ->
          ~H"""
          <div class="text-red-500">
            <%= @value %>
          </div>
          """
        end
      },
      # The ends_at field of our event schema. This field should not be sortable.
      ends_at: %{
        module: Backpex.Fields.Date,
        label: "Ends at"
      },
      # The published field of our url schema. We use the boolean field to display a switch button on edit views.
      published: %{
        module: Backpex.Fields.Boolean,
        label: "Published",
        sortable: false
      }
    ]
  end
end

  
    
  
  Templates


You are able to customize certain parts of Backpex. While you may use our app shell layout only you may also define functions to provide additional templates to be rendered on the resource LiveView or completely overwrite certain parts like the header or main content.
See render_resource_slot/3 for supported positions.
Example:
# in your resource configuration file

# to add content above main on index view
def render_resource_slot(assigns, :index, :before_main), do: ~H"Hello World!"

  
    
  
  Item Query


It is possible to manipulate the query when fetching resources for index, show and edit view by defining an item_query function.
In all queries we define a from query with a named binding to fetch all existing resources on index view or a specific resource on show / edit view.
After that, we call the item_query function. By default this returns the incoming query.
The item_query function makes it easy to add custom query expressions.
For example, you could filter posts by a published boolean on index view.
# in your resource configuration file

@impl Backpex.LiveResource
def item_query(query, :index, _assigns) do
query
|> where([post], post.published)
end
In this example we also made use of the named binding. It's always the name of the provided schema in snake_case.
It is recommended to build your item_query on top of the incoming query. Otherwise you will likely get binding errors.

  
    
  
  Authorize Actions


Use can?(_assigns, _action, _item) function in you resource configuration to limit access to item actions
(Actions: :index, :new, :show, :edit, :delete, :your_item_action_key, :your_resource_action_key).
The function is not required and returns true by default.
The item is nil for any action that does not require an item to be performed (:index, :new, :your_resource_action_key).
Examples:
# _item is nil for any action that does not require an item to be performed
def can?(_assigns, :new, _item), do: false

def can?(_assigns, :my_item_action, item), do: item.role == :admin

def can?(assigns, :my_resource_action, nil), do: assigns.current_user == :admin
Note that item actions are always displayed if they are defined. If you want to remove item actions completely, you must restrict access to them with can?/3 and remove the action with the item_actions/1 function.


  
    
  
  Resource Actions


You may define actions for certain resources in order to integrate complex processes into Backpex.
Action routes are automatically generated when using the live_resources macro.
For example you could add an invite process to your user resource as shown in the following.
defmodule MyAppWeb.Admin.Actions.Invite do
  use Backpex.ResourceAction

  import Ecto.Changeset

  @impl Backpex.ResourceAction
  def label, do: "Invite"

  @impl Backpex.ResourceAction
  def title, do: "Invite user"

  # you can reuse Backpex fields in the field definition
  @impl Backpex.ResourceAction
  def fields do
    [
      email: %{
        module: Backpex.Fields.Text,
        label: "Email",
        type: :string
      }
    ]
  end

  @required_fields ~w[email]a

  @impl Backpex.ResourceAction
  def changeset(change, attrs) do
    change
    |> cast(attrs, @required_fields)
    |> validate_required(@required_fields)
    |> validate_email(:email)
  end

  # your action to be performed
  @impl Backpex.ResourceAction
  def handle(_socket, params) do
    # Send mail

    # We suppose there was no error.
    if true do
      {:ok, "An email to #{params[:email]} was sent successfully."}
    else
      {:error, "An error occurred while sending an email to  #{params[:email]}!"}
    end
  end
end
# in your resource configuration file

# each action consists out of an unique id and the corresponding action module
@impl Backpex.LiveResource
def resource_actions() do
[
  %{
    module: MyWebApp.Admin.ResourceActions.Invite,
    id: :invite
  }
]
end

  
    
  
  Ordering


You may provide an init_order option to specify how the initial index page is being ordered.
# in your resource configuration file

use Backpex.LiveResource,
  ...,
  init_order: %{by: :inserted_at, direction: :desc}

  # Routing

  
    
  
  Routing


You are required to configure your router in order to point to the resources created in before steps.
Make sure to use the Backpex.InitAssigns hook to ensure all Backpex assigns are applied to the LiveViews.
You have to use the Backpex.Router.live_resources/3 macro to generate routes for your resources.
# MyAppWeb.Router

import Backpex.Router

scope "/admin", MyAppWeb do
pipe_through :browser

live_session :default, on_mount: Backpex.InitAssigns do
  live_resources("/events", EventLive)
end
In addition you have to use the Backpex.Router.backpex_routes macro. It will add some more routes at base scope. You can place this anywhere in your router.
We will mainly use this routes to insert a Backpex.CookieController. We need it in order to save some user related settings (e.g. which columns on index view you selected to be active).
# MyAppWeb.Router

import Backpex.Router

scope "/" do
  pipe_through :browser

  backpex_routes()
end

  
    
  
  Searching


You may flag fields as searchable. A search input will appear automatically on the resource index view.
# in your resource configuration file
@impl Backpex.LiveResource
def fields do
  [
    %{
      ...,
      searchable: true
    }
  ]
end
For a custom placeholder, you can use the elixir search_placeholder/0 callback.
# in your resource configuration file
@impl Backpex.LiveResource
def search_placeholder, do: "This will be shown in the search input."
In addition to basic searching, Backpex allows you to perform full-text searches on resources (see Full-Text Search Guide).

  
    
  
  Hooks


You may define hooks that are called after their respective action. Those hooks are on_item_created, on_item_updated and on_item_deleted.
These methods receive the socket and the corresponding item and are expected to return a socket.
# in your resource configuration file
@impl Backpex.LiveResource
def on_item_created(socket, item) do
  # send an email on user creation
  socket
end

  
    
  
  PubSub


PubSub settings are required in order to support live updates.
# in your resource configuration file
use Backpex.LiveResource,
    ...,
    pubsub: Demo.PubSub, # PubSub name of the project.
    topic: "events", # The topic for PubSub
    event_prefix: "event_" # The event prefix for Pubsub, to differentiate between events of different resources when subscribed to multiple resources
In addition you may react to ...deleted, ...updated and ...created events via handle_info
# in your resource configuration file
@impl Phoenix.LiveView
def handle_info({"event_created", item}, socket) do
  # make something in response to the event, for example show a toast to all users currently on the resource that an event has been created.
  {:noreply, socket}
end

  
    
  
  Navigation


You may define a custom navigation path that is called after the item is saved.
The method receives the socket, the live action and the corresponding item and is expected to return a route path.
# in your resource configuration file
@impl Backpex.LiveResource
def return_to(socket, assigns, _action, _item) do
  # return to user index after saving
  Routes.user_path(socket, :index)
end

  
    
  
  Panels


You are able to define panels to group certain fields together. Panels are displayed in the provided order.
The Backpex.LiveResource.panels/0 function has to return a keyword list with an identifier and label for each panel.
You can move fields into panels with the panel field configuration that has to return the identifier of the corresponding panel. Fields without a panel are displayed in the :default panel. The :default panel has no label.
Note that a panel is not displayed when there are no fields in it.

# in your resource configuration file
@impl Backpex.LiveResource
def panels do
  [
    contact: "Contact"
  ]
end

# in your fields list
@impl Backpex.LiveResource
def fields do
  [
    %{
      ...,
      panel: :contact
    }
  ]
end

  
    
  
  Default values


It is possible to assign default values to fields.
# in your resource configuration file
@impl Backpex.LiveResource
def fields do
  [
    username: %{
      default: fn _assigns -> "Default Username" end
    }
  ]
end
Note that default values are set when creating new resources only.


  
    
  
  Alignment


You may align fields on index view. By default fields are aligned to the left.
We currently support the following alignments: :left, :center and :right.
# in your resource configuration file
@impl Backpex.LiveResource
def fields do
  [
    %{
      ...,
      align: :center
    }
  ]
end
In addition to field alignment, you can align the labels on form views (index, edit, resource_action) using the align_label field option.
We currently support the following label orientations: :top, :center and :bottom.
# in your resource configuration file
@impl Backpex.LiveResource
def fields do
  [
    %{
      ...,
      align_label: :top
    }
  ]
end

  
    
  
  Fields Visibility


You are able to change visibility of fields based on certain conditions (assigns).
Imagine you want to implement a checkbox in order to toggle an input field (post likes). Initially, the input field should be visible when it has a certain value (post likes > 0).
# in your resource configuration file
@impl Backpex.LiveResource
def fields do
  [
    # show_likes is a virtual field in the post schema
    show_likes: %{
      module: Backpex.Fields.Boolean,
      label: "Show likes",
      # initialize the button based on the likes value
      select: dynamic([post: p], fragment("? > 0", p.likes)),
    },
    likes: %{
      module: Backpex.Fields.Number,
      label: "Likes",
      # display the field based on the `show_likes` value
      # the value can be part of the changeset or item (when edit view is opened initially).
      visible: fn
        %{live_action: :new} = assigns ->
          Map.get(assigns.changeset.changes, :show_likes)

        %{live_action: :edit} = assigns ->
          Map.get(assigns.changeset.changes, :show_likes, Map.get(assigns.item, :show_likes, false))

        _assigns ->
          true
      end
    }
  ]
end
Note that hidden fields are not exempt from validation by Backpex itself and the visible function is not executed on :index.

In addition to visible/1, we provide a can?1 function that you can use to determine the visibility of a field.
It can also be used on :index and takes the assigns as a parameter.
# in your resource configuration file
inserted_at: %{
  module: Backpex.Fields.DateTime,
  label: "Created At",
  can?: fn
    %{live_action: :show} = _assigns ->
      true

    _assigns ->
      false
  end
}

  
    
  
  Tooltips


We support tooltips via daisyUI.

  
    
  
  Index Editable


A small number of fields support index editable. These fields can be edited inline on the index view.
You must enable index editable for a field.
# in your resource configuration file
def fields do
  [
    name: %{
      module: Backpex.Fields.Text,
      label: "Name",
      index_editable: true
    }
  ]
end
Currently supported by the following fields:
	Backpex.Fields.BelongsTo
	Backpex.Fields.Date
	Backpex.Fields.DateTime
	Backpex.Fields.Number
	Backpex.Fields.Select
	Backpex.Fields.Text

Note you can add index editable support to your custom fields by defining the render_index_form/1 function and enabling index editable for your field.


  
    
  
  Additional classes for index table rows


We provide the Backpex.LiveResource.index_row_class option to add additional classes to table rows
on the index view. This allows you, for example, to color the rows.
# in your resource configuration file
@impl Backpex.LiveResource
def index_row_class(assigns, item, selected, index), do: "bg-yellow-100"
Note that we call the function twice. Once for the row on the tr element and a second time for the item action overlay, because in most cases the overlay should have the same style applied.
For this reason, Tailwind CSS modifiers such as even and odd will not always work as intended. Use the provided index instead. The index starts with 0 for the first item.


      


      
        Summary


  
    Callbacks
  


    
      
        
  
    
    Backpex.Resource - Backpex v0.3.0

Generic context module for Backpex resources.

Backpex.ResourceAction behaviour
    



      
Behaviour implemented by all resource action modules.

  
    
  
  Example


defmodule MyAppWeb.Actions.Invite do
  use Backpex.ResourceAction

  import Ecto.Changeset

  @impl Backpex.ResourceAction
  def title, do: "Invite user"

  @impl Backpex.ResourceAction
  def label, do: "Invite"

  @impl Backpex.ResourceAction
  def fields do
    [
      email: %{
        module: Backpex.Fields.Text,
        label: "Email",
        type: :string
      }
    ]
  end

  @impl Backpex.ResourceAction
  def changeset(change, attrs) do
    change
    |> cast(attrs, [:email])
    |> validate_required([:email]))
  end

  @impl Backpex.ResourceAction
  def handle(_socket, params) do
    # Send mail

    # Success
    {:ok, "An email to #{params[:email]} was sent successfully."}

    # Failure
    {:error, "An error occurred while sending an email to #{params[:email]}!"}
  end
end
use Backpex.ResourceAction
When you use Backpex.ResourceAction, the Backpex.ResourceAction module will set @behavior Backpex.ResourceAction.
In addition it will implement the Backpex.ResourceAction.init_change function in order to generate a schemaless changeset by default.

Provides LiveView routing for Backpex resources.

BackpexWeb 
    



      
The entrypoint for defining the web interface of Backpex.

  
    
  
  Example


use BackpexWeb, :html

Contains common HTML functions.

Contains all Backpex form components.

Backpex.HTML.Layout 
    



      
Contains all Backpex layout components.

Contains all Backpex resource components.

Backpex.Field behaviour
    



      
Behaviour implemented by all fields.
A field defines how a column is rendered on index, show and edit views. In the resource configuration file you can configure
a list of fields. You may create your own field by implementing this behaviour. A field has to be a LiveComponent.
When creating your own field, you can use the field macro from the BackpexWeb module. It automatically implements the Backpex.Field behaviour
and defines some aliases and imports.
The simplest version of a custom field would look like this:
use BackpexWeb, :field

@impl Backpex.Field
def render_value(assigns) do
  ~H"""
  <p>
    <%= HTML.pretty_value(@value) %>
  </p>
  """
end

@impl Backpex.Field
def render_form(assigns) do
  ~H"""
  <div>
    <Layout.field_container>
      <:label>
        <Layout.input_label text={@field_options[:label]} />
      </:label>
      <BackpexForm.field_input type="text" form={@form} field_name={@name} field_options={@field_options} />
    </Layout.field_container>
  </div>
  """
end
The render_value/1 function returns markup that is used to display a value on index and show views.
The render_form/1 function returns markup that is used to render a form on edit and new views.
The list of fields in the resource configuration has to be a keyword list. The key has to be the name of the column.
The value has to be a list of options represented as a map. At least you are required to provide the module of the field and a label as options.
For extra information and options you may have to look into the corresponding field documentation.

  
    
  
  Example


def fields do
  [
    rating: %{
      module: Backpex.Fields.Text,
      label: "Rating"
    }
  ]
end

  
    
  
  Read-only fields


Fields can be configured to be read-only. In edit view, these fields are rendered with the additional HTML attributes readonly and disabled,
ensuring that the user cannot interact with the field or change its value.
In index view, if read-only and index editable are set to true, forms will be rendered with the readonly HTML attribute.
Currently read-only configuration is possible for Backpex.Fields.Text and Backpex.Fields.Textarea on edit view.
On the index view, read-only is supported for all fields with the index editable option.
You can also add read-only functionality to a custom field. To do this, you need to define a render_form_readonly/1 function.
This function must return markup to be used when read-only is enabled.
@impl Backpex.Field
def render_form_readonly(assigns) do
  ~H"""
  <div>
    <Layout.field_container>
      <:label>
        <Layout.input_label text={@field[:label]} />
      </:label>
      <BackpexForm.field_input
        type="text"
        form={@form}
        field_name={@name}
        field_options={@field_options}
        readonly
        disabled
      />
    </Layout.field_container>
  </div>
  """
end
When defining a custom field with index editable support, you need to handle the read-only state in the index editable markup.
We pass a readonly value to index fields, which will be true or false depending on the read-only option of the field and the can?/3 function.
Fields can be set to read-only in the configuration map of the field by adding the readonly option. This key must
contain either a boolean value or a function that returns a boolean value.
The function takes assigns as a parameter. This allows the field to be set as read-only programmatically, as the
following example illustrates:
rating: %{
  module: Backpex.Fields.Text,
  label: "Rating",
  readonly: fn assigns ->
    assigns.current_user.role in [:employee]
  end
}

  
    
  
  Custom Alias Configuration


Backpex automatically generates aliases for queries in your fields. However, if you try to add two BelongsTo fields of the same association, you will encounter an error indicating that the alias is already in use by another field. To resolve this issue, Backpex allows the assignment of custom aliases to fields, eliminating naming conflicts in queries.
To use a custom alias, especially with a :select query, define it within your field configuration. This allows you to reference associated data via the custom alias.
Example usage of a custom alias together with a :select:
second_category: %{
  module: Backpex.Fields.BelongsTo,
  label: "Second Category",
  display_field: :name,
  searchable: true,
  custom_alias: :second_category,
  select: dynamic([second_category: sc], sc.name)
},

  
    
  
  Computed Fields


Sometimes you want to compute new fields based on existing fields.
Imagine there is a user table with first_name and last_name. Now, on your index view you want to add a column to display
the full_name. You could create a generated column in you database, but there are several reasons for not adding generated
columns for all computed fields you want to display in your application.
Backpex adds a way to support this.
There is a select option you may add to a field. This option has to return a dynamic. This query will then be executed to
select fields when listing your resources. In addition this query will also be used to order / search this field.
Therefore you can display the full_name of your users by adding the following field to the resource configuration file.
full_name: %{
  module: Backpex.Fields.Text,
  label: "Full Name",
  searchable: true,
  except: [:edit],
  select: dynamic([user: u], fragment("concat(?, ' ', ?)", u.first_name, u.last_name))
}
We are using a database fragment to build the full_name based on the first_name and last_name of an user. Backpex will select this field when listing resources automatically.
Ordering and searching works the same like on all other fields, because Backpex uses the query you provided in the dynamic in order / search queries, too.
We recommend to display this field on index and show view only.
Note: You are required to add a virtual field full_name to your user schema. Otherwise, Backpex is not able to select this field.

Computed fields also work in associations.
For example, you are able to add a select query to a BelongsTo field.
Imagine you want to display a list of posts with the corresponding authors (users). The user column should be a full_name computed by the first_name and last_name:
user: %{
  module: Backpex.Fields.BelongsTo,
  label: "Full Name",
  display_field: :full_name,
  select: dynamic([user: u], fragment("concat(?, ' ', ?)", u.first_name, u.last_name)),
  options_query: fn query, _assigns ->
    query |> select_merge([user], %{full_name: fragment("concat(?, ' ', ?)", user.first_name, user.last_name)})
  end
}
We recommend to add a select_merge to the options_query where you select the same field. Otherwise, displaying the same values in the select form on edit page will not work.
Do not forget to add the virtual field full_name to your user schema in this example, too.

  
    
  
  Error Customisation


Each field can define a translate_error/1 function to use custom error messages. The function is called for each error and must return a tuple with a message and metadata.
For example, if you want to indicate that an integer input must be a number:
number: %{
  module: Backpex.Fields.Number,
  label: "Number",
  translate_error: fn
    {_msg, [type: :integer, validation: :cast] = metadata} = _error ->
      {"has to be a number", metadata}

    error ->
      error
  end
}

Backpex.Fields.BelongsTo 
    



      
A field for handling a belongs_to relation.

  
    
  
  Options


	:display_field - The field of the relation to be used for searching, ordering and displaying values.
	:display_field_form - Optional field to be used to display form values.
	:live_resource - The live resource of the association. Used to generate links navigating to the association.
	:options_query - Manipulates the list of available options in the select.
Defaults to fn (query, _field) -> query end which returns all entries.
	:prompt - The text to be displayed when no option is selected or function that receives the assigns.


  
    
  
  Example


@impl Backpex.LiveResource
def fields do
[
  user: %{
    module: Backpex.Fields.BelongsTo,
    label: "Username",
    display_field: :username,
    options_query: &where(&1, [user], user.role == :admin),
    live_resource: DemoWeb.UserLive
  }
]
end

Backpex.Fields.Boolean 
    



      
A field for handling a boolean value.

  
    
  
  Options


	:debounce - Optional integer timeout value (in milliseconds), "blur" or function that receives the assigns.
	:throttle - Optional integer timeout value (in milliseconds) or function that receives the assigns.

Backpex.Fields.Currency 
    



      
A field for handling a currency value.

  
    
  
  Options


	:debounce - Optional integer timeout value (in milliseconds), "blur" or function that receives the assigns.
	:throttle - Optional integer timeout value (in milliseconds) or function that receives the assigns.


  
    
  
  Schema


Backpex.Ecto.Amount.Type provides a type for Ecto to store a amount. The underlying data type should be an integer.
For a full list of configuration options see: https://hexdocs.pm/money/Money.html#module-configuration
schema "article" do
  field :price, Backpex.Ecto.Amount.Type
  ...
end

schema "article" do
  field :price, Backpex.Ecto.Amount.Type, currency: :EUR, opts: [separator: ".", delimiter: ","]
  ...
end

  
    
  
  Example


@impl Backpex.LiveResource
def fields do
  [
    price: %{
      module: Backpex.Fields.Currency,
      label: "Price"
    }
  ]
end

Backpex.Fields.Date 
    



      
A field for handling a date value.

  
    
  
  Options


	:format - Defines the date format printed on the index view.
Defaults to %Y-%m-%d.
	:debounce - Optional integer timeout value (in milliseconds), "blur" or function that receives the assigns.
	:throttle - Optional integer timeout value (in milliseconds) or function that receives the assigns.


  
    
  
  Example


@impl Backpex.LiveResource
def fields do
  [
    created_at: %{
      module: Backpex.Fields.Date,
      label: "Created At",
      format: "%d.%m.%Y"
    }
  ]
end

Backpex.Fields.DateTime 
    



      
A field for handling a date time value.

  
    
  
  Options


	:format - Format string which will be used to format the date time value.
Defaults to %Y-%m-%d %I:%M %p
	:debounce - Optional integer timeout value (in milliseconds), "blur" or function that receives the assigns.
	:throttle - Optional integer timeout value (in milliseconds) or function that receives the assigns.


  
    
  
  Example


@impl Backpex.LiveResource
def fields do
  [
    created_at: %{
      module: Backpex.Fields.DateTime,
      label: "Created At",
      format: "%Y.%m.%d %I:%M %p" # optional
    }
  ]
end

Backpex.Fields.HasMany 
    



      
A field for handling a has_many relation.
This field can not be orderable or searchable.

  
    
  
  Options


	:display_field - The field of the relation to be used for searching, ordering and displaying values.
	:display_field_form - Optional field to be used to display form values.
	:live_resource - The live resource of the association. Used to generate links navigating to the associations.
	:options_query - Manipulates the list of available options in the multi select.
Defaults to fn (query, _field) -> query end which returns all entries.
	:prompt - The text to be displayed when no options are selected or function that receives the assigns.
Defaults to "Select options...".
	:not_found_text - The text to be displayed when no options are found.
Defaults to "No options found".
	:live_resource - Optional live resource module used to generate links to the corresponding show view for the item.
	:query_limit - Optional limit passed to the query to fetch new items. Set to nil to have no limit.
Defaults to 10.


  
    
  
  Example


@impl Backpex.LiveResource
def fields do
[
  posts: %{
    module: Backpex.Fields.HasMany,
    label: "Posts",
    display_field: :title,
    options_query: &where(&1, [user], user.role == :admin),
    live_resource: DemoWeb.PostLive
  }
]
end

Backpex.Fields.HasManyThrough 
    



      
A field for handling a has_many (through) relation.
This field is not orderable or searchable.
Warning
This field is in beta state. Use at your own risk.


  
    
  
  Options


	:display_field - The field of the relation to be used for displaying options in the select.
	:live_resource - The corresponding live resource of the association. Used to display the title of the modal and generate defaults for :child_fields fields.
	:sort_by - A list of columns by which the child element output will be sorted. The sorting takes place in ascending order.
	:child_fields - WIP
	:pivot_fields - List to map additional data of the pivot table to Backpex fields.
	:options_query - Manipulates the list of available options in the select. Can be used to select additional data for the display_field option or to limit the available entries.
Defaults to fn (query, _field) -> query end which returns all entries.


  
    
  
  Example


@impl Backpex.LiveResource
def fields do
[
  addresses: %{
    module: Backpex.Fields.HasManyThrough,
    label: "Addresses",
    display_field: :street,
    live_resource: DemoWeb.AddressLive,
    sort_by: [:zip, :city],
    pivot_fields: [
      type: %{
        module: Backpex.Fields.Select,
        label: "Address Type",
        options: [Shipping: "shipping", Billing: "billing"]
      }
    ]
  }
]
end
The field requires a Ecto.Schema.has_many/3 relation with a mandatory through option in the main schema. Any extra column in the pivot table besides the relational id's must be mapped in the pivot_fields option or given a default value.

Backpex.Fields.InlineCRUD 
    



      
A field to handle inline CRUD operations. It can be used with either an embeds_many or has_many (association) type column.

  
    
  
  Options


	:type - The type of the field. Either :embed or :assoc.
	:child_fields - A list of input fields to be used. Currently only support Backpex.Fields.Text fields.

You can add additional classes to child field inputs by setting the class option in the list of child_fields.
The class can be a string or a function that takes the assigns and must return a string.
In addition, you can optionally specify the input type of child field inputs with the input_type option. We currently
support :text and :textarea. The input_type defaults to :text.
Important
Everything is currently handled by plain text input.


  
    
  
  EmbedsMany


The field in the migration must be of type :map. You also need to use ecto's cast_embed/2 in the changeset.

  
    
  
  Example


def changeset(your_schema, attrs) do
  your_schema
  ...
  |> cast_embed(:your_field,
    with: &your_field_changeset/2,
    sort_param: :your_field_order,
    drop_param: :your_field_delete
  )
  ...
end
Important
We use the Ecto :sort_param and :drop_param to keep track of order and dropped items. Therefore, you need to use these options as well in your changeset. The name has to be <field_name>_order and <field_name>_delete.


  
    
  
  HasMany (Association)


A HasMany relation does not require any special configuration. You can simply define a basic Ecto.Schema.has_many/3 relation to be used with the Backpex.Fields.InlineCRUD field.
Important
You need to set on_replace: :delete to be able to delete items, and on_delete: :delete_all to be able to delete a resource with existing items. It is recommended that you also add on_delete: :delete_all to your migration.


  
    
  
  Example


@impl Backpex.LiveResource
def fields do
  [
    embeds_many: %{
      module: Backpex.Fields.InlineCRUD,
      label: "EmbedsMany",
      type: :embed,
      except: [:index],
      child_fields: [
        field1: %{
          module: Backpex.Fields.Text,
          label: "Label1"
        },
        field2: %{
          module: Backpex.Fields.Text,
          label: "Label2"
        }
      ]
    }
  ]
end

Backpex.Fields.ManyToMany 
    



      
A field for handling a many_to_many relation.
This field can not be orderable or searchable.

  
    
  
  Options


	:display_field - The field of the relation to be used for searching, ordering and displaying values.
	:display_field_form - Optional field to be used to display form values.
	:live_resource - The live resource of the association. Used to generate links navigating to the associations.
	:options_query - Manipulates the list of available options in the multi select.
Defaults to fn (query, _field) -> query end which returns all entries.
	:prompt - The text to be displayed when no options are selected or function that receives the assigns.
Defaults to "Select options...".
	:not_found_text - The text to be displayed when no options are found.
Defaults to "No options found".
	:live_resource - Optional live resource module used to generate links to the corresponding show view for the item.
	:query_limit - Optional limit passed to the query to fetch new items. Set to nil to have no limit.
Defaults to 10.


  
    
  
  Example


@impl Backpex.LiveResource
def fields do
[
  posts: %{
    module: Backpex.Fields.HasMany,
    label: "Posts",
    display_field: :title,
    options_query: &where(&1, [user], user.role == :admin),
    live_resource: DemoWeb.PostLive
  }
]
end

Backpex.Fields.MultiSelect 
    



      
A field for handling a multi select with predefined options.
This field can not be searchable.

  
    
  
  Options


	:options - Required (keyword) list of options to be used for the select.
	:prompt - The text to be displayed when no options are selected or function that receives the assigns.
Defaults to "Select options...".
	:not_found_text - The text to be displayed when no options are found.
Defaults to "No options found".


  
    
  
  Example


@impl Backpex.LiveResource
def fields do
  [
    users: %{
      module: Backpex.Fields.MultiSelect,
      label: "Users",
      options: fn _assigns -> [{"Alex", "user_id_alex"}, {"Bob", "user_id_bob"}] end
    },
  ]

Backpex.Fields.Number 
    



      
A field for handling a number value.

  
    
  
  Options


	:placeholder - Optional placeholder value or function that receives the assigns.
	:debounce - Optional integer timeout value (in milliseconds), "blur" or function that receives the assigns.
	:throttle - Optional integer timeout value (in milliseconds) or function that receives the assigns.


      




  

  
    
    Backpex.Fields.Select - Backpex v0.3.0
    
    

    


  
  

    
Backpex.Fields.Select 
    



      
A field for handling a select value.

  
    
  
  Options


	:options - Required (keyword) list of options or function that receives the assigns.
	:prompt - The text to be displayed when no option is selected or function that receives the assigns.


  
    
  
  Example


@impl Backpex.LiveResource
def fields do
  [
    role: %{
      module: Backpex.Fields.Select,
      label: "Role",
      options: [Admin: admin, User: user]
    }
  ]
end

      




  

  
    
    Backpex.Fields.Text - Backpex v0.3.0
    
    

    


  
  

    
Backpex.Fields.Text 
    



      
A field for handling a text value.

  
    
  
  Options


	:placeholder - Optional placeholder value or function that receives the assigns.
	:debounce - Optional integer timeout value (in milliseconds), "blur" or function that receives the assigns.
	:throttle - Optional integer timeout value (in milliseconds) or function that receives the assigns.


      




  

  
    
    Backpex.Fields.Textarea - Backpex v0.3.0
    
    

    


  
  

    
Backpex.Fields.Textarea 
    



      
A field for handling long text values.

  
    
  
  Options


	:placeholder - Optional placeholder value or function that receives the assigns.
	:debounce - Optional integer timeout value (in milliseconds), "blur" or function that receives the assigns.
	:throttle - Optional integer timeout value (in milliseconds) or function that receives the assigns.


      




  

  
    
    Backpex.Fields.URL - Backpex v0.3.0
    
    

    


  
  

    
Backpex.Fields.URL 
    



      
A field for handling an URL value.

  
    
  
  Options


	:placeholder - Optional placeholder value or function that receives the assigns.
	:debounce - Optional integer timeout value (in milliseconds), "blur" or function that receives the assigns.
	:throttle - Optional integer timeout value (in milliseconds) or function that receives the assigns.


      




  

  
    
    Backpex.Fields.Upload - Backpex v0.3.0
    
    

    


  
  

    
Backpex.Fields.Upload 
    



      
A field for handling uploads.
Warning
This field does not currently support Phoenix.LiveView.UploadWriter and direct / external uploads.


  
    
  
  Options


	:upload_key (atom) - Required identifier for the upload field (the name of the upload).
	:accept (list) - Required filetypes that will be accepted.
	:max_entries (integer) - Required number of max files that can be uploaded.
	:max_file_size (integer) - Optional maximum file size in bytes to be allowed to uploaded. Defaults 8 MB (8_000_000).
	:list_existing_files (function) - Required function that returns a list of all uploaded files based on an item.
	:file_label (function) - Optional function to get the label of a single file.
	:consume_upload (function) - Required function to consume file uploads.
	:put_upload_change (function) - Required function to add file paths to the change.
	:remove_uploads (function) - Required function that is being called after saving an item to be able to delete removed files

Info
The following examples copy uploads to a static folder in the application. In a production environment, you should consider uploading files to an appropriate object store.


  
    
  
  Options in detail


The upload_key, accept, max_entries and max_file_size options are forwarded to https://hexdocs.pm/phoenix_live_view/Phoenix.LiveView.html#allow_upload/3. See the documentation for more information.

  
    
  
  list_existing_files


Parameters
	:socket - The socket.
	:item (struct) - The item without its changes.

The function is being used to display existing uploads. The function receives the socket and the item and has to return a list of strings. Removed files during an edit of an item are automatically removed from the list. This option is required.
Example
def list_existing_files(_socket, item), do: item.files

  
    
  
  file_label


Parameters
	:file (string) - The file.

The function can be used to modify a file label based on a file. In the following example each file will have an "_upload" suffix. This option is optional.
Example
def file_label(file), do: file <> "_upload"

  
    
  
  consume_upload


Parameters
	:socket - The socket.
	:item (struct) - The item without its changes.
	:meta - The upload meta.
	:entry - The upload entry.

The function is used to consume uploads. It is called after the item has been saved and is used to copy the files to a specific destination. Backpex will use this function as a callback for consume_uploaded_entries. See https://hexdocs.pm/phoenix_live_view/uploads.html#consume-uploaded-entries for more details. This option is required.
Example
defp consume_upload(_socket, _item, %{path: path} = _meta, entry) do
  file_name = ...
  file_url = ...
  static_dir = ...
  dest = Path.join([:code.priv_dir(:demo), "static", static_dir, file_name])

  File.cp!(path, dest)

  {:ok, file_url}
end

  
    
  
  put_upload_change


Parameters
	:socket - The socket.
	:change (map) - The current change / attrs that will be passed to the changeset function.
	:item (struct) - The item without its changes. On create will this will be an empty map.
	uploaded_entries (tuple) - The completed and in progress entries for the upload.
	removed_entries (list) - A list of removed uploads during edit.
	action (atom) - The action (:validate or :insert)

This function is used to modify the change based on certain parameters. It is important because it ensures that file paths are added to the item change and therefore persisted in the database. This option is required.
Example
def put_upload_change(_socket, change, item, uploaded_entries, removed_entries, action) do
  existing_files = item.files -- removed_entries

  new_entries =
    case action do
      :validate ->
        elem(uploaded_entries, 1)

      :insert ->
        elem(uploaded_entries, 0)
    end

  files = existing_files ++ Enum.map(new_entries, fn entry -> file_name(entry) end)

  Map.put(change, "images", files)
end

  
    
  
  remove_uploads


Parameters
	:socket - The socket.
	:item (struct) - The item without its changes.
	removed_entries (list) - A list of removed uploads during edit.

Example
defp remove_uploads(_socket, _item, removed_entries) do
  for file <- removed_entries do
    file_path = ...
    File.rm!(file_path)
  end
end

  
    
  
  Full Single File Example


In this example we are adding an avatar upload for a user. We implement it so that exactly one avatar must exist.
defmodule Demo.Repo.Migrations.AddAvatarToUsers do
  use Ecto.Migration

  def change do
    alter table(:users) do
      add(:avatar, :string, null: false, default: "")
    end
  end
end

defmodule Demo.User do
  use Ecto.Schema

  schema "users" do
    field(:avatar, :string, default: "")
    ...
  end

  def changeset(user, attrs, _metadata \\ []) do
    user
    |> cast(attrs, [:avatar])
    |> validate_required([:avatar])
    |> validate_change(:avatar, fn
      :avatar, "too_many_files" ->
        [avatar: "has to be exactly one"]

      :avatar, "" ->
        [avatar: "can't be blank"]

      :avatar, _avatar ->
        []
    end)
  end
end

defmodule DemoWeb.UserLive do
  use Backpex.LiveResource,
    ...

  @impl Backpex.LiveResource
  def fields do
    [
      avatar: %{
        module: Backpex.Fields.Upload,
        label: "Avatar",
        upload_key: :avatar,
        accept: ~w(.jpg .jpeg .png),
        max_entries: 1,
        max_file_size: 512_000,
        put_upload_change: &put_upload_change/6,
        consume_upload: &consume_upload/4,
        remove_uploads: &remove_uploads/3,
        list_existing_files: &list_existing_files/1,
        render: fn
          %{value: value} = assigns when value == "" or is_nil(value) ->
            ~H"<p><%= Backpex.HTML.pretty_value(@value) %></p>"

          assigns ->
            ~H'<img class="h-10 w-auto" src={file_url(@value)} />'
        end
      },
      ...
    ]
  end

  defp list_existing_files(%{avatar: avatar} = _item) when avatar != "" and not is_nil(avatar), do: [avatar]
  defp list_existing_files(_item), do: []

  def put_upload_change(_socket, change, item, uploaded_entries, removed_entries, action) do
    existing_files = list_existing_files(item) -- removed_entries

    new_entries =
      case action do
        :validate ->
          elem(uploaded_entries, 1)

        :insert ->
          elem(uploaded_entries, 0)
      end

    files = existing_files ++ Enum.map(new_entries, fn entry -> file_name(entry) end)

    case files do
      [file] ->
        Map.put(change, "avatar", file)

      [_file | _other_files] ->
        Map.put(change, "avatar", "too_many_files")

      [] ->
        Map.put(change, "avatar", "")
    end
  end

  defp consume_upload(_socket, _item, %{path: path} = _meta, entry) do
    file_name = file_name(entry)
    dest = Path.join([:code.priv_dir(:demo), "static", upload_dir(), file_name])

    File.cp!(path, dest)

    {:ok, file_url(file_name)}
  end

  defp remove_uploads(_socket, _item, removed_entries) do
    for file <- removed_entries do
      path = Path.join([:code.priv_dir(:demo), "static", upload_dir(), file])
      File.rm!(path)
    end
  end

  defp file_url(file_name) do
    static_path = Path.join([upload_dir(), file_name])
    Phoenix.VerifiedRoutes.static_url(DemoWeb.Endpoint, "/" <> static_path)
  end

  defp file_name(entry) do
    [ext | _] = MIME.extensions(entry.client_type)
    "#{entry.uuid}.#{ext}"
  end

  defp upload_dir, do: Path.join(["uploads", "user", "avatar"])
end

  
    
  
  Full Multi File Example


In this example, we are adding images to a product resource. We limit the images to a maximum of 2.
defmodule Demo.Repo.Migrations.AddImagesToProducts do
  use Ecto.Migration

  def change do
    alter table(:products) do
      dd(:images, {:array, :string})
    end
  end
end

defmodule Demo.Product do
  use Ecto.Schema

  schema "products" do
    field(:images, {:array, :string})
    ...
  end

  def changeset(user, attrs, _metadata \\ []) do
    user
    |> cast(attrs, [:images])
    |> validate_length(:images, max: 2)
  end
end

defmodule DemoWeb.ProductLive do
  use Backpex.LiveResource,
    ...

  @impl Backpex.LiveResource
  def fields do
    [
      images: %{
        module: Backpex.Fields.Upload,
        label: "Images",
        upload_key: :images,
        accept: ~w(.jpg .jpeg .png),
        max_entries: 2,
        max_file_size: 512_000,
        put_upload_change: &put_upload_change/6,
        consume_upload: &consume_upload/4,
        remove_uploads: &remove_uploads/3,
        list_existing_files: &list_existing_files/1,
        render: fn
          %{value: value} = assigns when is_list(value) ->
            ~H'''
            <div>
              <img :for={img <- @value} class="h-10 w-auto" src={file_url(img)} />
            </div>
            '''

          assigns ->
            ~H'<p><%= Backpex.HTML.pretty_value(@value) %></p>'
        end,
        except: [:index, :resource_action],
        align: :center
      },
      ...
    ]
  end

  defp list_existing_files(%{images: images} = _item) when is_list(images), do: images
  defp list_existing_files(_item), do: []

  defp put_upload_change(_socket, change, item, uploaded_entries, removed_entries, action) do
    existing_files = list_existing_files(item) -- removed_entries

    new_entries =
      case action do
        :validate ->
          elem(uploaded_entries, 1)

        :insert ->
          elem(uploaded_entries, 0)
      end

    files = existing_files ++ Enum.map(new_entries, fn entry -> file_name(entry) end)

    Map.put(change, "images", files)
  end

  defp consume_upload(_socket, _item, %{path: path} = _meta, entry) do
    file_name = file_name(entry)
    dest = Path.join([:code.priv_dir(:demo), "static", upload_dir(), file_name])

    File.cp!(path, dest)

    {:ok, file_url(file_name)}
  end

  defp remove_uploads(_socket, _item, removed_entries) do
    for file <- removed_entries do
      path = Path.join([:code.priv_dir(:demo), "static", upload_dir(), file])
      File.rm!(path)
    end
  end

  defp file_url(file_name) do
    static_path = Path.join([upload_dir(), file_name])
    Phoenix.VerifiedRoutes.static_url(DemoWeb.Endpoint, "/" <> static_path)
  end

  defp file_name(entry) do
    [e